In this Maple file we compute the Hamiltonian evolutions for the second element of the Painlevé 1 hierarchy and we obtain the Lax matrices in the geometric gauge in terms of the (q_i,p_i) Darboux coordinates. Formulas are compared with the theoretical ones.

Notation: The general deformation operator is \mathcal{L}=\hbar (alpha7*\partial_{t_{\infty^{(1)},7} +alpha6*\partial_{t_{\infty^{(1)},6}+... alpha1*\partial_{t_{\infty^{(1)},1})

```
> restart:
  with(LinearAlgebra):
  rinfty:=4:
  g:=rinfty-2:
  tinfty27:=-tinfty17:
  tinfty25:=-tinfty15:
  tinfty23:=-tinfty13:
  tinfty21:=-tinfty11:
  tinfty20:=-tinfty10:
  tinfty26:=tinfty16:
  tinfty24:=tinfty14:
  tinfty22:=tinfty12:
  tinfty10:=0:
  Pinfty01 := -tinfty12;
  Pinfty11 := -tinfty14;
  Pinfty21 := -tinfty16;
  Pinfty22 := -(1/2)*tinfty11*tinfty17+(1/2)*tinfty12*tinfty16-
  (1/2)*tinfty13*tinfty15+(1/4)*tinfty14^2;
  Pinfty32 := -(1/2)*tinfty13*tinfty17+(1/2)*tinfty14*tinfty16-
  (1/4)*tinfty15^2;
  Pinfty42 := -(1/2)*tinfty15*tinfty17+(1/4)*tinfty16^2;
  Pinfty52 := -(1/4)*tinfty17^2;

  P1:=x-> Pinfty01+Pinfty11*x+Pinfty21*x^2;
  P2:=x-> Pinfty02+Pinfty12*x+Pinfty22*x^2+Pinfty32*x^3+Pinfty42*
  x^4+Pinfty52*x^5;
  tdP2:=unapply(Pinfty22*x^2+Pinfty32*x^3+Pinfty42*x^4+Pinfty52*
  x^5,x);

  dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
  dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):
  dtdP2dlambda:=unapply(diff(tdP2(lambda),lambda),lambda):

  c3:=-(7*alpha6*tinfty17-6*alpha7*tinfty16)/(42*tinfty17):
  c2:=-(35*alpha4*tinfty17^2-28*alpha5*tinfty16*tinfty17-20*alpha7*
  tinfty14*tinfty17+20*alpha7*tinfty15*tinfty16)/(140*tinfty17^2):
  c1:=-(105*alpha2*tinfty17^3-70*alpha3*tinfty16*tinfty17^2-42*
  alpha5*tinfty14*tinfty17^2+42*alpha5*tinfty15*tinfty16*tinfty17
```

```
-30*alpha7*tinfty12*tinfty17^2+30*alpha7*tinfty13*tinfty16*
tinfty17+30*alpha7*tinfty14*tinfty15*tinfty17-30*alpha7*
tinfty15^2*tinfty16)/(210*tinfty17^3):

nuMoins1:=2*alpha7/(7*tinfty17):
nu0:=(2*(7*alpha5*tinfty17-5*alpha7*tinfty15))/(35*tinfty17^2):
nu1:=-(2*(tinfty13*tinfty17-tinfty15^2))*alpha7/(7*tinfty17^3)-2*
tinfty15*alpha5/(5*tinfty17^2)+2*alpha3/(3*tinfty17):
nu2:=-(2*(tinfty11*tinfty17^2-2*tinfty13*tinfty15*tinfty17+
tinfty15^3))*alpha7/(7*tinfty17^4)-(2*(tinfty13*tinfty17-
tinfty15^2))*alpha5/(5*tinfty17^3)-2*tinfty15*alpha3/(3*
tinfty17^2)+2*alpha1/tinfty17:

mu1:=2*(-35*alpha3*q2*tinfty17^3+21*alpha5*q2*tinfty15*
tinfty17^2+15*alpha7*q2*tinfty13*tinfty17^2-15*alpha7*q2*
tinfty15^2*tinfty17+105*alpha1*tinfty17^3-35*alpha3*tinfty15*
tinfty17^2-21*alpha5*tinfty13*tinfty17^2+21*alpha5*tinfty15^2*
tinfty17-15*alpha7*tinfty11*tinfty17^2+30*alpha7*tinfty13*
tinfty15*tinfty17-15*alpha7*tinfty15^3)/(105*tinfty17^4*(q1-q2)):
mu2:=-1/(105*tinfty17^4*(q1-q2))* 2*(-35*alpha3*q1*tinfty17^3+21*
alpha5*q1*tinfty15*tinfty17^2+15*alpha7*q1*tinfty13*tinfty17^2
-15*alpha7*q1*tinfty15^2*tinfty17+105*alpha1*tinfty17^3-35*
alpha3*tinfty15*tinfty17^2-21*alpha5*tinfty13*tinfty17^2+21*
alpha5*tinfty15^2*tinfty17-15*alpha7*tinfty11*tinfty17^2+30*
alpha7*tinfty13*tinfty15*tinfty17-15*alpha7*tinfty15^3):

dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):
L:=Matrix(2,2,0):
L[1,1]:=0:
L[1,2]:=1:
L[2,1]:=-tdP2(lambda)+C1*lambda+C0 - h*p1/(lambda-q1)-h*p2/
(lambda-q2):
L[2,2]:= P1(lambda)  +h/(lambda-q1)+h/(lambda-q2) :

A:=Matrix(2,2,0):
A[1,1]:=c3*lambda^3+c2*lambda^2+c1*lambda+c0+ rho1/(lambda-q1)+
rho2/(lambda-q2):
A[1,2]:=nuMoins1*lambda+nu0+ mu1/(lambda-q1)+ mu2/(lambda-q2):
A[2,1]:=AA21(lambda):
A[2,2]:=AA22(lambda):
dAdlambda:=Matrix(2,2,0):
```

```
for i from 1 to 2 do for j from 1 to 2 do dAdlambda[i,j]:=diff(A
[i,j],lambda):  od: od:

L:
A:

Q2:=unapply( -p1*(lambda-q2)/(q1-q2)-p2*(lambda-q1)/(q2-q1),
lambda);
simplify(Q2(q1));
simplify(Q2(q2));

J:=Matrix(2,2,0):
J[1,1]:=1:
J[1,2]:=0:
J[2,1]:=Q2(lambda)/(lambda-q1)/(lambda-q2):
J[2,2]:=1/(lambda-q1)/(lambda-q2):
dJdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dJdlambda[i,j]:=diff(J
[i,j],lambda): od: od:
J:

Lcheck:=simplify(Multiply(Multiply(J,L),J^(-1))+h*Multiply
(dJdlambda,J^(-1))):

simplify(Multiply(Multiply(J,A),J^(-1))[1,1]-A[1,1]+Q2(lambda)*A
[1,2]);
```

$$Pinfty01 := -tinfty12$$
$$Pinfty11 := -tinfty14$$
$$Pinfty21 := -tinfty16$$
$$Pinfty22 := -\frac{tinfty11\,tinfty17}{2} + \frac{tinfty12\,tinfty16}{2} - \frac{tinfty13\,tinfty15}{2} + \frac{tinfty14^2}{4}$$
$$Pinfty32 := -\frac{tinfty13\,tinfty17}{2} + \frac{tinfty14\,tinfty16}{2} - \frac{tinfty15^2}{4}$$
$$Pinfty42 := -\frac{tinfty15\,tinfty17}{2} + \frac{tinfty16^2}{4}$$
$$Pinfty52 := -\frac{tinfty17^2}{4}$$
$$P1 := x \mapsto Pinfty01 + Pinfty11\,x + Pinfty21\,x^2$$
$$P2 := x \mapsto Pinfty02 + Pinfty12\,x + Pinfty22\,x^2 + Pinfty32\,x^3 + Pinfty42\,x^4 + Pinfty52\,x^5$$
$$tdP2 := x \mapsto -\frac{x^5\,tinfty17^2}{4} + x^4\left(-\frac{tinfty15\,tinfty17}{2} + \frac{tinfty16^2}{4}\right) + x^3\left(-\frac{tinfty13\,tinfty17}{2} + \frac{tinfty14\,tinfty16}{2} - \frac{tinfty15^2}{4}\right) + x^2\left(-\frac{tinfty11\,tinfty17}{2} + \frac{tinfty12\,tinfty16}{2}\right.$$

$$-\frac{tinfty13\, tinfty15}{2}+\frac{tinfty14^2}{4}\Big)$$

$$Q2 := \lambda \mapsto -\frac{p1\,(\lambda-q2)}{q1-q2}-\frac{p2\,(\lambda-q1)}{q2-q1}$$

$$-p1$$

$$-p2$$

$$0 \quad \textbf{(1)}$$

The compatibility equation reads \mathcal{L}L=h\partial_x A+[A,L]. Because the first line of L is trivial (0,1) we may obtain A[2,1] and A[2,2] to complete the knowledge of A

```
> CurlyL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):

  Entry11:=CurlyL[1,1]:
  Entry12:=CurlyL[1,2]:

  AA21:=unapply(solve(Entry11=0,AA21(lambda)),lambda):
  AA21bis:=h*dAdlambda[1,1]+A[1,2]*L[2,1]:

  simplify(AA21(lambda)-AA21bis);
  AA22:=unapply(solve(Entry12=0,AA22(lambda)),lambda):
  AA22bis:=h*dAdlambda[1,2]+A[1,1]+A[1,2]*L[2,2]:

  simplify(AA22(lambda)-AA22bis);
  simplify(Entry11);
  simplify(Entry12);
  CurlyL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
```

$$0$$

$$0$$

$$0$$

$$0 \quad \textbf{(2)}$$

Let us now compute the action of \mathcal{L} on L[2,2] and L[2,1]

```
> Entry22:=simplify(CurlyL[2,2]):
  Entry22TermLambdaMoinsq1Cube:=factor(residue(Entry22*(lambda-q1)
  ^2,lambda=q1));
  Entry22TermLambdaMoinsq1Square:=factor(residue(Entry22*(lambda-
  q1),lambda=q1)):
  Entry22TermLambdaMoinsq1:=factor(residue(Entry22,lambda=q1)):

  Entry22TermLambdaMoinsq2Cube:=factor(residue(Entry22*(lambda-q2)
  ^2,lambda=q2));
  Entry22TermLambdaMoinsq2Square:=factor(residue(Entry22*(lambda-
  q2),lambda=q2)):
  Entry22TermLambdaMoinsq2:=factor(residue(Entry22,lambda=q2)):

  Entry22TermLambdaInfty4:=factor(-residue(Entry22/lambda^5,lambda=
```

```
infinity));
Entry22TermLambdaInfty3:=factor(-residue(Entry22/lambda^4,lambda=
infinity));
Entry22TermLambdaInfty2:=factor(-residue(Entry22/lambda^3,lambda=
infinity));
Entry22TermLambdaInfty1:=factor(-residue(Entry22/lambda^2,lambda=
infinity));
Entry22TermLambdaInfty0:=factor(-residue(Entry22/lambda,lambda=
infinity));

simplify( Entry22-(Entry22TermLambdaMoinsq1Square/(lambda-q1)^2+
Entry22TermLambdaMoinsq1/(lambda-q1)
+Entry22TermLambdaMoinsq2Square/(lambda-q2)^2+
Entry22TermLambdaMoinsq2/(lambda-q2)
+Entry22TermLambdaInfty0+Entry22TermLambdaInfty1*lambda+
Entry22TermLambdaInfty2*lambda^2+Entry22TermLambdaInfty3*
lambda^3+Entry22TermLambdaInfty4*lambda^4) );
L[2,2];
```

$$Entry22TermLambdaMoinsq1Cube := 0$$
$$Entry22TermLambdaMoinsq2Cube := 0$$
$$Entry22TermLambdaInfty4 := 0$$
$$Entry22TermLambdaInfty3 := 0$$
$$Entry22TermLambdaInfty2 := -\alpha6\, h$$
$$Entry22TermLambdaInfty1 := -\alpha4\, h$$
$$Entry22TermLambdaInfty0 := -\alpha2\, h$$
$$0$$
$$-tinfty16\,\lambda^2 - tinfty14\,\lambda - tinfty12 + \frac{h}{\lambda - q1} + \frac{h}{\lambda - q2} \quad \textbf{(3)}$$

Because the deformation operator is \mathcal{L}=\hbar (alpha7*\partial_{t_{\infty^{(1)},7} +alpha6*\partial_{t_{\infty^{(1)},6}+... alpha1*\partial_{t_{\infty^{(1)},1}), we get

```
> L22OrderLambda3:=-residue(L[2,2]/lambda^4,lambda=infinity);
L22OrderLambda2:=-residue(L[2,2]/lambda^3,lambda=infinity);
L22OrderLambda1:=-residue(L[2,2]/lambda^2,lambda=infinity);
L22OrderLambda0:=-residue(L[2,2]/lambda^1,lambda=infinity);
Equation1:=factor(simplify(h*(alpha7*diff(L22OrderLambda2,
tinfty17)+alpha6*diff(L22OrderLambda2,tinfty16)+alpha5*diff
(L22OrderLambda2,tinfty15)+alpha4*diff(L22OrderLambda2,tinfty14)+
alpha3*diff(L22OrderLambda2,tinfty13)+alpha2*diff
(L22OrderLambda2,tinfty12)+alpha1*diff(L22OrderLambda2,tinfty11))
- Entry22TermLambdaInfty2));
Equation2:=factor(simplify(h*(alpha7*diff(L22OrderLambda1,
tinfty17)+alpha6*diff(L22OrderLambda1,tinfty16)+alpha5*diff
(L22OrderLambda1,tinfty15)+alpha4*diff(L22OrderLambda1,tinfty14)
```

```
+alpha3*diff(L22OrderLambda1,tinfty13)+alpha2*diff
(L22OrderLambda1,tinfty12)+alpha1*diff(L22OrderLambda1,tinfty11))
- Entry22TermLambdaInfty1));
Equation3:=factor(simplify(h*(alpha7*diff(L22OrderLambda0,
tinfty17)+alpha6*diff(L22OrderLambda0,tinfty16)+alpha5*diff
(L22OrderLambda0,tinfty15)+alpha4*diff(L22OrderLambda0,tinfty14)
+alpha3*diff(L22OrderLambda0,tinfty13)+alpha2*diff
(L22OrderLambda0,tinfty12)+alpha1*diff(L22OrderLambda0,tinfty11))
- Entry22TermLambdaInfty0));
```

$$L22OrderLambda3 := 0$$
$$L22OrderLambda2 := -tinfty16$$
$$L22OrderLambda1 := -tinfty14$$
$$L22OrderLambda0 := -tinfty12$$
$$Equation1 := 0$$
$$Equation2 := 0$$
$$Equation3 := 0 \qquad (4)$$

We have verified that the computations of c_{\infty,1}, c_{\infty,2} and c_{\infty,3} are correct. Let us now look at the evolution of q_1 and q_2

```
> CurlyLq1:=factor(Entry22TermLambdaMoinsq1Square/h):
  CurlyLq2:=factor(Entry22TermLambdaMoinsq2Square/h):
```

Let us now look at \mathcal{L}[L[2,1]]

```
> Entry21:=simplify(CurlyL[2,1]):
  Entry21TermLambdaMoinsq1Cube:=factor(residue(Entry21*(lambda-q1)
  ^2,lambda=q1));
  Entry21TermLambdaMoinsq1Square:=factor(residue(Entry21*(lambda-
  q1),lambda=q1)):
  Entry21TermLambdaMoinsq1:=factor(residue(Entry21,lambda=q1)):

  Entry21TermLambdaMoinsq2Cube:=factor(residue(Entry21*(lambda-q2)
  ^2,lambda=q2));
  Entry21TermLambdaMoinsq2Square:=factor(residue(Entry21*(lambda-
  q2),lambda=q2)):
  Entry21TermLambdaMoinsq2:=factor(residue(Entry21,lambda=q2)):

  Entry21TermLambdaInfty8:=factor(-residue(Entry21/lambda^9,lambda=
  infinity));
  Entry21TermLambdaInfty7:=factor(-residue(Entry21/lambda^8,lambda=
  infinity));
  Entry21TermLambdaInfty6:=factor(-residue(Entry21/lambda^7,lambda=
  infinity));
  Entry21TermLambdaInfty5:=factor(-residue(Entry21/lambda^6,lambda=
  infinity));
  Entry21TermLambdaInfty4:=factor(-residue(Entry21/lambda^5,lambda=
```

```
infinity)):
Entry21TermLambdaInfty3:=factor(-residue(Entry21/lambda^4,lambda=
infinity)):
Entry21TermLambdaInfty2:=factor(-residue(Entry21/lambda^3,lambda=
infinity)):
Entry21TermLambdaInfty1:=factor(-residue(Entry21/lambda^2,lambda=
infinity)):
Entry21TermLambdaInfty0:=factor(-residue(Entry21/lambda,lambda=
infinity)):

simplify( Entry21-(Entry21TermLambdaMoinsq1Cube/(lambda-q1)^3+
Entry21TermLambdaMoinsq1Square/(lambda-q1)^2+
Entry21TermLambdaMoinsq1/(lambda-q1)
+Entry21TermLambdaMoinsq2Cube/(lambda-q2)^3+
Entry21TermLambdaMoinsq2Square/(lambda-q2)^2+
Entry21TermLambdaMoinsq2/(lambda-q2)
+Entry21TermLambdaInfty0+Entry21TermLambdaInfty1*lambda+
Entry21TermLambdaInfty2*lambda^2+Entry21TermLambdaInfty3*lambda^3
+Entry21TermLambdaInfty4*lambda^4+Entry21TermLambdaInfty5*
lambda^5+Entry21TermLambdaInfty6*lambda^6
+Entry21TermLambdaInfty7*lambda^7
) );
L[2,1];
```

$$Entry21TermLambdaMoinsq1Cube := \frac{1}{35\,(q1-q2)\,tinfty17^4}\Big(\big(-70\,\alpha 3\,p1\,q2\,tinfty17^3 + 42\,\alpha 5\,p1\,q2\,tinfty15\,tinfty17^2 + 30\,\alpha 7\,p1\,q2\,tinfty13\,tinfty17^2 - 30\,\alpha 7\,p1\,q2\,tinfty15^2\,tinfty17 + 105\,q1\,\rho 1\,tinfty17^4 - 105\,q2\,\rho 1\,tinfty17^4 + 210\,tinfty17^3\,p1\,\alpha 1 - 70\,tinfty17^2\,tinfty15\,p1\,\alpha 3 - 42\,tinfty17^2\,tinfty13\,p1\,\alpha 5 + 42\,tinfty17\,tinfty15^2\,p1\,\alpha 5 - 30\,tinfty17^2\,tinfty11\,p1\,\alpha 7 + 60\,tinfty17\,tinfty15\,tinfty13\,p1\,\alpha 7 - 30\,tinfty15^3\,p1\,\alpha 7\big)\,h^2\Big)$$

$$Entry21TermLambdaMoinsq2Cube := -\frac{1}{35\,(q1-q2)\,tinfty17^4}\Big(\big(-70\,\alpha 3\,p2\,q1\,tinfty17^3 + 42\,\alpha 5\,p2\,q1\,tinfty15\,tinfty17^2 + 30\,\alpha 7\,p2\,q1\,tinfty13\,tinfty17^2 - 30\,\alpha 7\,p2\,q1\,tinfty15^2\,tinfty17 - 105\,q1\,\rho 2\,tinfty17^4 + 105\,q2\,\rho 2\,tinfty17^4 + 210\,tinfty17^3\,p2\,\alpha 1 - 70\,tinfty17^2\,tinfty15\,p2\,\alpha 3 - 42\,tinfty17^2\,tinfty13\,p2\,\alpha 5 + 42\,tinfty17\,tinfty15^2\,p2\,\alpha 5 - 30\,tinfty17^2\,tinfty11\,p2\,\alpha 7 + 60\,tinfty17\,tinfty15\,tinfty13\,p2\,\alpha 7 - 30\,tinfty15^3\,p2\,\alpha 7\big)\,h^2\Big)$$

$$Entry21TermLambdaInfty8 := 0$$
$$Entry21TermLambdaInfty7 := 0$$
$$Entry21TermLambdaInfty6 := 0$$

$$Entry21TermLambdaInfty5 := \frac{\alpha 7\ tinfty17\ h}{2}$$

$$0$$

$$\frac{tinfty17^2 \lambda^5}{4} - \left(-\frac{tinfty15\ tinfty17}{2} + \frac{tinfty16^2}{4} \right) \lambda^4 - \left(-\frac{tinfty13\ tinfty17}{2} + \frac{tinfty14\ tinfty16}{2} - \frac{tinfty15^2}{4} \right) \lambda^3 - \left(-\frac{tinfty11\ tinfty17}{2} + \frac{tinfty12\ tinfty16}{2} - \frac{tinfty13\ tinfty15}{2} + \frac{tinfty14^2}{4} \right) \lambda^2 + C1\ \lambda + C0 - \frac{h\ p1}{\lambda - q1} - \frac{h\ p2}{\lambda - q2} \quad \textbf{(5)}$$

We check that rho_j=-p_j*mu_j cancel the cubic pole at lambda=q_j

```
> rho1:=-p1*mu1:
rho2:=-p2*mu2:
simplify(Entry21TermLambdaMoinsq1Cube);
simplify(Entry21TermLambdaMoinsq2Cube);
```

$$0$$

$$0 \quad \textbf{(6)}$$

We check that the value of \td{P}_2 are compatible with the system

```
> L21OrderLambda5:=-residue(L[2,1]/lambda^6,lambda=infinity);
L21OrderLambda4:=-residue(L[2,1]/lambda^5,lambda=infinity);
L21OrderLambda3:=-residue(L[2,1]/lambda^4,lambda=infinity);
L21OrderLambda2:=-residue(L[2,1]/lambda^3,lambda=infinity);
L21OrderLambda1:=-residue(L[2,1]/lambda^2,lambda=infinity);
L21OrderLambda0:=-residue(L[2,1]/lambda^1,lambda=infinity);
Equation1:=factor(simplify(h*(alpha7*diff(L21OrderLambda5,
tinfty17)+alpha6*diff(L21OrderLambda5,tinfty16)+alpha5*diff
(L21OrderLambda5,tinfty15)+alpha4*diff(L21OrderLambda5,tinfty14)+
alpha3*diff(L21OrderLambda5,tinfty13)+alpha2*diff
(L21OrderLambda5,tinfty12)+alpha1*diff(L21OrderLambda5,tinfty11))
- Entry21TermLambdaInfty5));
Equation2:=factor(simplify(h*(alpha7*diff(L21OrderLambda4,
tinfty17)+alpha6*diff(L21OrderLambda4,tinfty16)+alpha5*diff
(L21OrderLambda4,tinfty15)+alpha4*diff(L21OrderLambda4,tinfty14)+
alpha3*diff(L21OrderLambda4,tinfty13)+alpha2*diff
(L21OrderLambda4,tinfty12)+alpha1*diff(L21OrderLambda4,tinfty11))
- Entry21TermLambdaInfty4));
Equation3:=factor(simplify(h*(alpha7*diff(L21OrderLambda3,
tinfty17)+alpha6*diff(L21OrderLambda3,tinfty16)+alpha5*diff
(L21OrderLambda3,tinfty15)+alpha4*diff(L21OrderLambda3,tinfty14)
+alpha3*diff(L21OrderLambda3,tinfty13)+alpha2*diff
(L21OrderLambda3,tinfty12)+alpha1*diff(L21OrderLambda3,tinfty11))
- Entry21TermLambdaInfty3));
Equation4:=factor(simplify(h*(alpha7*diff(L21OrderLambda2,
tinfty17)+alpha6*diff(L21OrderLambda2,tinfty16)+alpha5*diff
```

```
(L21OrderLambda2,tinfty15)+alpha4*diff(L21OrderLambda2,tinfty14)+
alpha3*diff(L21OrderLambda2,tinfty13)+alpha2*diff
(L21OrderLambda2,tinfty12)+alpha1*diff(L21OrderLambda2,tinfty11))
- Entry21TermLambdaInfty2));
```

$$L21OrderLambda5 := \frac{tinfty17^2}{4}$$
$$L21OrderLambda4 := \frac{tinfty15\ tinfty17}{2} - \frac{tinfty16^2}{4}$$
$$L21OrderLambda3 := \frac{tinfty13\ tinfty17}{2} - \frac{tinfty14\ tinfty16}{2} + \frac{tinfty15^2}{4}$$
$$L21OrderLambda2 := \frac{tinfty11\ tinfty17}{2} - \frac{tinfty12\ tinfty16}{2} + \frac{tinfty13\ tinfty15}{2} - \frac{tinfty14^2}{4}$$
$$L21OrderLambda1 := C1$$
$$L21OrderLambda0 := C0$$
$$Equation1 := 0$$
$$Equation2 := 0$$
$$Equation3 := 0$$
$$Equation4 := 0 \quad (7)$$

```
> CurlyLp1Fonction:=unapply(-Entry21TermLambdaMoinsq1/h,C0,C1):
  CurlyLp2Fonction:=unapply(-Entry21TermLambdaMoinsq2/h,C0,C1):
> Equation5:=simplify(Entry21TermLambdaMoinsq1Square-(-p1*h*
  CurlyLq1)):
  Equation6:=simplify(Entry21TermLambdaMoinsq2Square-(-p2*h*
  CurlyLq2)):
> C0:=-(-q1^5*q2*tinfty17^2+q1*q2^5*tinfty17^2-2*q1^4*q2*tinfty15*
  tinfty17+q1^4*q2*tinfty16^2+2*q1*q2^4*tinfty15*tinfty17-q1*q2^4*
  tinfty16^2-2*q1^3*q2*tinfty13*tinfty17+2*q1^3*q2*tinfty14*
  tinfty16-q1^3*q2*tinfty15^2+2*q1*q2^3*tinfty13*tinfty17-2*q1*
  q2^3*tinfty14*tinfty16+q1*q2^3*tinfty15^2+4*p1*q1^2*q2*tinfty16
  -4*p2*q1*q2^2*tinfty16-2*q1^2*q2*tinfty11*tinfty17+2*q1^2*q2*
  tinfty12*tinfty16-2*q1^2*q2*tinfty13*tinfty15+q1^2*q2*
  tinfty14^2+2*q1*q2^2*tinfty11*tinfty17-2*q1*q2^2*tinfty12*
  tinfty16+2*q1*q2^2*tinfty13*tinfty15-q1*q2^2*tinfty14^2+4*p1*q1*
  q2*tinfty14-4*p2*q1*q2*tinfty14+4*p1^2*q2+4*p1*q2*tinfty12-4*
  p2^2*q1-4*p2*q1*tinfty12+4*h*p1-4*h*p2)/(4*(q1-q2)):
  C1:=1/4/(q1-q2)*(-q1^5*tinfty17^2+q2^5*tinfty17^2-2*q1^4*
  tinfty15*tinfty17+q1^4*tinfty16^2+2*q2^4*tinfty15*tinfty17-q2^4*
  tinfty16^2-2*q1^3*tinfty13*tinfty17+2*q1^3*tinfty14*tinfty16-
  q1^3*tinfty15^2+2*q2^3*tinfty13*tinfty17-2*q2^3*tinfty14*
  tinfty16+q2^3*tinfty15^2+4*p1*q1^2*tinfty16-4*p2*q2^2*tinfty16-2*
  q1^2*tinfty11*tinfty17+2*q1^2*tinfty12*tinfty16-2*q1^2*tinfty13*
  tinfty15+q1^2*tinfty14^2+2*q2^2*tinfty11*tinfty17-2*q2^2*
  tinfty12*tinfty16+2*q2^2*tinfty13*tinfty15-q2^2*tinfty14^2+4*p1*
```

```
q1*tinfty14-4*p2*q2*tinfty14+4*p1^2+4*p1*tinfty12-4*p2^2-4*p2*
tinfty12):
simplify(Equation5);
simplify(Equation6);
```

$$0$$
$$0 \quad (8)$$

We check that C_0 and C_1 are in accordance with the theoretical ones

```
> Vinfty:=Matrix(g,g,0):
Vinfty[1,1]:=1:
Vinfty[1,2]:=1:
Vinfty[2,1]:=q1:
Vinfty[2,2]:=q2:
Vinfty;
RHSC:=Matrix(2,2,0):
RHSC[1,1]:=p1^2-P1(q1)*p1+tdP2(q1)+h*(p2-p1)/(q1-q2):
RHSC[2,1]:=p2^2-P1(q2)*p2+tdP2(q2)+h*(p1-p2)/(q2-q1):
CVectorTheo:=Multiply((Transpose(Vinfty))^(-1),RHSC):
C0Theo:=CVectorTheo[1,1]:
C1Theo:=CVectorTheo[2,1]:
simplify(C0-C0Theo);
simplify(C1-C1Theo);
```

$$\begin{bmatrix} 1 & 1 \\ q1 & q2 \end{bmatrix}$$
$$0$$
$$0 \quad (9)$$

We now compute the evolution of p_1 and p_2

```
> CurlyLp1:=factor(simplify(CurlyLp1Fonction(C0,C1))):
CurlyLp2:=factor(simplify(CurlyLp2Fonction(C0,C1))):
```

We check the evolutions of q_1,q_2,p_1,p_2 in accordance with our general formulas:

```
> CurlyLq1Theo:=2*mu1*(p1-1/2*P1(q1))-h*nu0-h*nuMoins1*q1-h*(mu1+
mu2)/(q1-q2):
CurlyLq2Theo:=2*mu2*(p2-1/2*P1(q2))-h*nu0-h*nuMoins1*q2-h*(mu2+
mu1)/(q2-q1):
simplify(CurlyLq1-CurlyLq1Theo);
simplify(CurlyLq2-CurlyLq2Theo);

CurlyLp1Theo:=h*(mu2+mu1)*(p2-p1)/(q1-q2)^2+mu1*(p1*dP1dlambda
(q1)-dtdP2dlambda(q1)+1*C1*q1^0) +h*nuMoins1*p1+ h*(1*c1*q1^0+2*
c2*q1^1+3*c3*q1^2):
CurlyLp2Theo:=h*(mu1+mu2)*(p1-p2)/(q2-q1)^2+mu2*(p2*dP1dlambda
(q2)-dtdP2dlambda(q2)+1*C1*q2^0) +h*nuMoins1*p2+ h*(1*c1*q2^0+2*
c2*q2^1+3*c3*q2^2):
```

```
series(factor(simplify(CurlyLp1-CurlyLp1Theo)),p1=0);
simplify(CurlyLp2-CurlyLp2Theo);
```

$$0$$
$$0$$
$$0$$
$$0 \quad (10)$$

We check the expression of the general Hamiltonian

```
> Hamiltonian:=-h/2*(mu1+mu2)*(p1-p2)/(q1-q2)-h/2*(mu2+mu1)*(p2-p1)
/(q2-q1)-h*(nu0*(p1+p2)+nuMoins1*(q1*p1+q2*p2))
+ mu1*(p1^2-p1*P1(q1)+tdP2(q1))+ mu2*(p2^2-p2*P1(q2)+tdP2(q2))-h*
(c1*q1^1+c2*q1^2+c3*q1^3+c1*q2^1+c2*q2^2+c3*q2^3):
simplify(CurlyLp1-(-diff(Hamiltonian,q1)));
simplify(CurlyLq1-(diff(Hamiltonian,p1)));
simplify(CurlyLp2-(-diff(Hamiltonian,q2)));
simplify(CurlyLq2-(diff(Hamiltonian,p2)));
```

$$0$$
$$0$$
$$0$$
$$0 \quad (11)$$

```
> factor(simplify(Hamiltonian-(nu1*C0+nu2*C1-h*nu0*(p1+p2)-h*
nuMoins1*(q1*p1+q2*p2)-h*c1*(q1+q2)-h*c2*(q1^2+q2^2)-h*c3*(q1^3+
q2^3) )));
```

$$0 \quad (12)$$

We check the first gauge transformation to remove apparent singularities

```
> Lcheck[1,1];
Lcheck[1,2];
Lcheck[2,2]:
ProductLambdaMinusq:=unapply((lambda-q1)*(lambda-q2),lambda):
Lcheck11Theo:=-Q2(lambda):
Lcheck12Theo:=(lambda-q1)*(lambda-q2):
Lcheck22Theo:=P1(lambda)+Q2(lambda):
Lcheck21Theo:=h*diff(Q2(lambda)/ProductLambdaMinusq(lambda),
lambda) +L[2,1]/ProductLambdaMinusq(lambda)-P1(lambda)*Q2(lambda)
/ProductLambdaMinusq(lambda)-(Q2(lambda))^2/ProductLambdaMinusq
(lambda):
simplify(Lcheck[1,1]-Lcheck11Theo);
simplify(Lcheck[1,2]-Lcheck12Theo);
simplify(Lcheck[2,2]-Lcheck22Theo);
simplify(Lcheck[2,1]-Lcheck21Theo);
```

$$\frac{(p1-p2)\,\lambda-p1\,q2+p2\,q1}{q1-q2}$$
$$(-\lambda+q1)\,(-\lambda+q2)$$
$$0$$

$$0$$
$$0$$
$$0 \qquad (13)$$

We now check the second gauge transformation

```
> G:=Matrix(2,2,0):
  G[1,1]:=1:
  G[2,2]:=1:
  G[2,1]:=1/2*tinfty16:
  G;
  dGdlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dGdlambda[i,j]:=diff(G
  [i,j],lambda): od: od:
  CurlyLG:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do CurlyLG[i,j]:=
  h*(alpha7*diff(G[i,j],tinfty17)+alpha6*diff(G[i,j],tinfty16)+
  alpha5*diff(G[i,j],tinfty15)+alpha4*diff(G[i,j],tinfty14)+alpha3*
  diff(G[i,j],tinfty13)+alpha2*diff(G[i,j],tinfty12)+alpha1*diff(G
  [i,j],tinfty11)): od: od:
  dGdlambda;
  CurlyLG;

  Ltilde:=simplify(Multiply(Multiply(G,Lcheck),G^(-1))+h*Multiply
  (dGdlambda,G^(-1))):
  Atilde:=simplify(Multiply(Multiply(G,Acheck),G^(-1))+Multiply
  (CurlyLG,G^(-1))):
```

$$\begin{bmatrix} 1 & 0 \\ \frac{tinfty16}{2} & 1 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 \\ \frac{\alpha 6\, h}{2} & 0 \end{bmatrix} \qquad (14)$$

```
> Ltilde11Orderlambda3:=factor(-residue(Ltilde[1,1]/lambda^4,
  lambda=infinity));
  Ltilde12Orderlambda3:=factor(-residue(Ltilde[1,2]/lambda^4,
  lambda=infinity));
  Ltilde21Orderlambda3:=factor(-residue(Ltilde[2,1]/lambda^4,
  lambda=infinity));
  Ltilde22Orderlambda3:=factor(-residue(Ltilde[2,2]/lambda^4,
  lambda=infinity));
```

$$Ltilde11Orderlambda3 := 0$$
$$Ltilde12Orderlambda3 := 0$$
$$Ltilde21Orderlambda3 := \frac{tinfty17^2}{4}$$
$$Ltilde22Orderlambda3 := 0 \quad (15)$$

```
> Ltilde11Orderlambda2:=factor(-residue(Ltilde[1,1]/lambda^3,
lambda=infinity));
Ltilde12Orderlambda2:=factor(-residue(Ltilde[1,2]/lambda^3,
lambda=infinity));
Ltilde21Orderlambda2:=factor(-residue(Ltilde[2,1]/lambda^3,
lambda=infinity));
Ltilde22Orderlambda2:=factor(-residue(Ltilde[2,2]/lambda^3,
lambda=infinity));
```

$$Ltilde11Orderlambda2 := -\frac{tinfty16}{2}$$
$$Ltilde12Orderlambda2 := 1$$
$$Ltilde21Orderlambda2 := \frac{tinfty17\,(q1\,tinfty17 + q2\,tinfty17 + 2\,tinfty15)}{4}$$
$$Ltilde22Orderlambda2 := -\frac{tinfty16}{2} \quad (16)$$

```
> Multiply(G,J);
```

$$\begin{bmatrix} 1 & 0 \\ \frac{tinfty16}{2} + \frac{-\frac{p1\,(\lambda - q2)}{q1 - q2} - \frac{p2\,(\lambda - q1)}{q2 - q1}}{(\lambda - q1)\,(\lambda - q2)} & \frac{1}{(\lambda - q1)\,(\lambda - q2)} \end{bmatrix} \quad (17)$$

```
> CurlyLJ:=Matrix(2,2,0):
CurlyLJ[1,1]:=0:
CurlyLJ[1,2]:=0:
CurlyLJ[2,2]:=diff(J[2,2],q1)*CurlyLq1+diff(J[2,2],p1)*
CurlyLp1+diff(J[2,2],q2)*CurlyLq2+diff(J[2,2],p2)*CurlyLp2
+h*(alpha7*diff(J[2,2],tinfty17)+alpha6*diff(J[2,2],tinfty16)+
alpha5*diff(J[2,2],tinfty15)+alpha4*diff(J[2,2],tinfty14)+alpha3*
diff(J[2,2],tinfty13)+alpha2*diff(J[2,2],tinfty12)+alpha1*diff(J
[2,2],tinfty11)) :
CurlyLJ[2,1]:=diff(J[2,1],q1)*CurlyLq1+diff(J[2,1],p1)*
CurlyLp1+diff(J[2,1],q2)*CurlyLq2+diff(J[2,1],p2)*CurlyLp2
+h*(alpha7*diff(J[2,1],tinfty17)+alpha6*diff(J[2,1],tinfty16)+
alpha5*diff(J[2,1],tinfty15)+alpha4*diff(J[2,1],tinfty14)+alpha3*
diff(J[2,1],tinfty13)+alpha2*diff(J[2,1],tinfty12)+alpha1*diff(J
[2,1],tinfty11)):
CurlyLJ:
Acheck:=simplify(Multiply(Multiply(J,A),J^(-1))+Multiply(CurlyLJ,
J^(-1))):
```

Expression in canonical coordinates after reduction

```
> tinfty17:=2:
  tinfty15:=0:
  tinfty16:=0:
  tinfty14:=0:
  tinfty12:=0:
  tinfty13:=2*tau1:
  tinfty11:=2*tau2:
```

```
> tdP2(lambda);

  C0bis:=-h*(p1-p2)/(q1-q2)+(q1*p2^2-q2*p1^2)/(q1-q2) +2*q1*q2*
  tau2+2*(q1+q2)*q1*q2*tau1+(q1+q2)*(q1^2+q2^2)*q1*q2;
  C1bis:=(p1^2-p2^2)/(q1-q2) -2*(q1^2+q1*q2+q2^2)*tau1 -2*(q1+q2)
  *tau2 -q1^4-q2*q1^3-q1^2*q2^2-q1*q2^3-q2^4;
  factor(simplify(series(C0-C0bis,tau2=0)));
  factor(simplify(series(C1-C1bis,tau2=0)));
```

$$-\lambda^5-2\lambda^3\tau 1-2\lambda^2\tau 2$$

$$C0bis := -\frac{h(p1-p2)}{q1-q2}+\frac{-q2\,p1^2+q1\,p2^2}{q1-q2}+2\,q1\,q2\,\tau 2+2\,(q1+q2)\,q1\,q2\,\tau 1+(q1+q2)\left(q1^2+q2^2\right)q1\,q2$$

$$C1bis := \frac{p1^2-p2^2}{q1-q2}-2\left(q1^2+q2\,q1+q2^2\right)\tau 1-2\,(q1+q2)\,\tau 2-q1^4-q2\,q1^3-q1^2\,q2^2-q1\,q2^3-q2^4$$

$$0$$

$$0 \qquad \textbf{(1.1)}$$

```
> mu1fonction:=unapply(mu1,
  alpha7,alpha6,alpha5,alpha4,alpha3,alpha2,alpha1);
  mu2fonction:=unapply(mu2,alpha7,alpha6,alpha5,alpha4,alpha3,
  alpha2,alpha1);
  CurlyLq1fonction:=unapply(CurlyLq1,alpha7,alpha6,alpha5,alpha4,
  alpha3,alpha2,alpha1):
  CurlyLp1fonction:=unapply(CurlyLp1,alpha7,alpha6,alpha5,alpha4,
  alpha3,alpha2,alpha1):
  CurlyLq2fonction:=unapply(CurlyLq2,alpha7,alpha6,alpha5,alpha4,
  alpha3,alpha2,alpha1):
  CurlyLp2fonction:=unapply(CurlyLp2,alpha7,alpha6,alpha5,alpha4,
  alpha3,alpha2,alpha1):
```

$$mu1fonction := (\alpha 7,\alpha 6,\alpha 5,\alpha 4,\alpha 3,\alpha 2,\alpha 1) \mapsto \frac{120\,\alpha 7\,q2\,\tau 1-280\,\alpha 3\,q2-168\,\alpha 5\,\tau 1-120\,\alpha 7\,\tau 2+840\,\alpha 1}{840\,(q1-q2)}$$

$$mu2fonction := (\alpha 7,\alpha 6,\alpha 5,\alpha 4,\alpha 3,\alpha 2,\alpha 1) \mapsto \qquad \textbf{(1.2)}$$

$$-\frac{120\,\alpha 7\,q1\,\tau 1-280\,\alpha 3\,q1-168\,\alpha 5\,\tau 1-120\,\alpha 7\,\tau 2+840\,\alpha 1}{840\,(q1-q2)}$$

```
> mu1tau1:=mu1fonction(0,0,0,0,2,0,0);
  mu1tau2:=mu1fonction(0,0,0,0,0,0,2);
```

$$mu1tau1:=-\frac{2\,q2}{3\,(q1-q2)}$$

$$mu1tau2:=\frac{2}{q1-q2} \tag{1.3}$$

```
> mu2tau1:=mu2fonction(0,0,0,0,2,0,0);
  mu2tau2:=mu2fonction(0,0,0,0,0,0,2);
```

$$mu2tau1:=\frac{2\,q1}{3\,(q1-q2)}$$

$$mu2tau2:=-\frac{2}{q1-q2} \tag{1.4}$$

```
> CurlyLq1tau1:=CurlyLq1fonction(0,0,0,0,2,0,0):
  CurlyLq1tau1bis:=-2*h/3/(q1-q2)-4/3*q2*p1/(q1-q2);
  simplify(series(CurlyLq1tau1-CurlyLq1tau1bis,h=0));
  CurlyLp1tau1:=CurlyLp1fonction(0,0,0,0,2,0,0):
  CurlyLp1tau1bis:=-2/3*(q2*p1^2-q2*p2^2)/(q1-q2)^2-2*h/3*(p1-p2)
  /(q1-q2)^2-2/3*(q2^4+2*q2^3*q1+3*q2^2*q1^2+4*q2*q1^3)  -4/3*q2*
  (q2+2*q1)*tau1-4/3*q2*tau2;
  simplify(series(CurlyLp1tau1-CurlyLp1tau1bis,p2=0));
```

$$CurlyLq1tau1bis:=-\frac{2\,h}{3\,(q1-q2)}-\frac{4\,p1\,q2}{3\,(q1-q2)}$$

$$0$$

$$CurlyLp1tau1bis:=-\frac{2\,(q2\,p1^2-p2^2\,q2)}{3\,(q1-q2)^2}-\frac{2\,h\,(p1-p2)}{3\,(q1-q2)^2}-\frac{8\,q2\,q1^3}{3}-2\,q1^2\,q2^2-\frac{4\,q1\,q2^3}{3}-\frac{2\,q2^4}{3}-\frac{4\,q2\,(q2+2\,q1)\,\tau 1}{3}-\frac{4\,q2\,\tau 2}{3}$$

$$0 \tag{1.5}$$

```
> CurlyLq2tau1:=CurlyLq2fonction(0,0,0,0,2,0,0):
  CurlyLq2tau1bis:=2*h/3/(q1-q2)+4/3*q1*p2/(q1-q2);
  simplify(series(CurlyLq2tau1-CurlyLq2tau1bis,h=0));
  CurlyLp2tau1:=CurlyLp2fonction(0,0,0,0,2,0,0):
  CurlyLp2tau1bis:=2/3*(q1*p1^2-q1*p2^2)/(q1-q2)^2+2*h/3*(p1-p2)/
  (q1-q2)^2-2/3*(q1^4+2*q1^3*q2+3*q1^2*q2^2+4*q1*q2^3)  -4/3*q1*
  (q1+2*q2)*tau1-4/3*q1*tau2;
  simplify(series(CurlyLp2tau1-CurlyLp2tau1bis,p2=0));
```

$$CurlyLq2tau1bis:=\frac{2\,h}{3\,(q1-q2)}+\frac{4\,p2\,q1}{3\,(q1-q2)}$$

$$0$$

$$CurlyLp2tau1bis:=\frac{2\,(p1^2\,q1-q1\,p2^2)}{3\,(q1-q2)^2}+\frac{2\,h\,(p1-p2)}{3\,(q1-q2)^2}-\frac{2\,q1^4}{3}-\frac{4\,q2\,q1^3}{3}$$

$$-2\,q1^2\,q2^2-\frac{8\,q1\,q2^3}{3}-\frac{4\,q1\,(q1+2\,q2)\,\tau 1}{3}-\frac{4\,q1\,\tau 2}{3}$$

$$0 \qquad \textbf{(1.6)}$$

```
> CurlyLq1tau2:=CurlyLq1fonction(0,0,0,0,0,0,2):
  CurlyLq1tau2bis:=4*p1/(q1-q2);
  simplify(series(CurlyLq1tau2-CurlyLq1tau2bis,h=0));
  CurlyLp1tau2:=CurlyLp1fonction(0,0,0,0,0,0,2):
  CurlyLp1tau2bis:=2*(p1^2-p2^2)/(q1-q2)^2+4*(2*q1+q2)*tau1+4*
  tau2+2*(4*q1^3+3*q1^2*q2+2*q1*q2^2+1*q2^3);
  simplify(series(CurlyLp1tau2-CurlyLp1tau2bis,tau1=0));
```

$$CurlyLq1tau2bis := \frac{4\,p1}{q1-q2}$$

$$0$$

$$CurlyLp1tau2bis := \frac{2\,(p1^2-p2^2)}{(q1-q2)^2}+4\,(q2+2\,q1)\,\tau 1+4\,\tau 2+8\,q1^3+6\,q1^2\,q2+4\,q1\,q2^2+2\,q2^3$$

$$0 \qquad \textbf{(1.7)}$$

```
> CurlyLq2tau2:=CurlyLq2fonction(0,0,0,0,0,0,2):
  CurlyLq2tau2bis:=-4*p2/(q1-q2);
  simplify(series(CurlyLq2tau2-CurlyLq2tau2bis,h=0));
  CurlyLp2tau2:=CurlyLp2fonction(0,0,0,0,0,0,2):
  CurlyLp2tau2bis:=-2*(p1^2-p2^2)/(q1-q2)^2+4*(2*q2+q1)*tau1+4*
  tau2+2*(4*q2^3+3*q2^2*q1+2*q2*q1^2+1*q1^3);
  simplify(series(CurlyLp2tau2-CurlyLp2tau2bis,tau1=0));
```

$$CurlyLq2tau2bis := -\frac{4\,p2}{q1-q2}$$

$$0$$

$$CurlyLp2tau2bis := -\frac{2\,(p1^2-p2^2)}{(q1-q2)^2}+4\,(q1+2\,q2)\,\tau 1+4\,\tau 2+2\,q1^3+4\,q1^2\,q2+6\,q1\,q2^2+8\,q2^3$$

$$0 \qquad \textbf{(1.8)}$$

```
> HamiltonianFonction:=unapply(simplify
  (Hamiltonian),alpha7,alpha6,alpha5,alpha4,alpha3,alpha2,alpha1)
  :
  H1:=HamiltonianFonction(0,0,0,0,2,0,0):
  H1bis:=2*(q1*p2^2-q2*p1^2)/3/(q1-q2)-2*h/3*(p1-p2)/(q1-q2)+2/3*
  (q1+q2)*(q1^2+q2^2)*q1*q2+4/3*(q1+q2)*q1*q2*tau1+4/3*q1*q2*
  tau2;
  simplify(series(H1-H1bis,tau2=0));

  H2:=HamiltonianFonction(0,0,0,0,0,0,2):
  H2bis:=2*(p1^2-p2^2)/(q1-q2)-2*(q1^4+q1^3*q2+q1^2*q2^2+q1*q2^3+
  q2^4)-4*(q1^2+q1*q2+q2^2)*tau1-4*tau2*(q1+q2);
```

```
simplify(series(H2-H2bis,p1=0));
```

$$H1bis := \frac{2\left(-q2\,p1^2+q1\,p2^2\right)}{3\,(q1-q2)}-\frac{2\,h\,(p1-p2)}{3\,(q1-q2)}+\frac{2\,(q1+q2)\left(q1^2+q2^2\right)q1\,q2}{3}+\frac{4\,(q1+q2)\,q1\,q2\,\tau 1}{3}+\frac{4\,q1\,q2\,\tau 2}{3}$$

$$0$$

$$H2bis := \frac{2\left(p1^2-p2^2\right)}{q1-q2}-2\,q1^4-2\,q2\,q1^3-2\,q1^2\,q2^2-2\,q1\,q2^3-2\,q2^4-4\left(q1^2+q2\,q1+q2^2\right)\tau 1-4\,(q1+q2)\,\tau 2$$

$$0 \qquad \textbf{(1.9)}$$

```
> simplify(Ltilde);
```

$$\left[\left[\frac{(p1-p2)\,\lambda-p1\,q2+p2\,q1}{q1-q2},\,(-\lambda+q1)\,(-\lambda+q2)\right],\right. \qquad \textbf{(1.10)}$$

$$\left[\frac{1}{(q1-q2)^2}\left(q1^5+(\lambda-q2)\,q1^4+\left(\lambda^2-q2\,\lambda+2\,\tau 1\right)q1^3+\left(\left(-\lambda^2-2\,\tau 1\right)q2+\lambda^3+2\,\lambda\,\tau 1+2\,\tau 2\right)q1^2-\left(q2^3+q2^2\,\lambda+\left(\lambda^2+2\,\tau 1\right)q2+2\,\lambda^3+4\,\lambda\,\tau 1+4\,\tau 2\right)q2\,q1+q2^5+\lambda\,q2^4+\left(\lambda^2+2\,\tau 1\right)q2^3+\left(\lambda^3+2\,\lambda\,\tau 1+2\,\tau 2\right)q2^2-(p1-p2)^2\right),\,\frac{(p2-p1)\,\lambda+p1\,q2-p2\,q1}{q1-q2}\right] \quad ,$$

```
> Ltilde21:=simplify(Ltilde[2,1]):
  Ltilde21bis:=lambda^3+(q1+q2)*lambda^2+(q1^2+q1*q2+q2^2+2*tau1)
  *lambda
  +q1^3+q1^2*q2+q1*q2^2+q2^3+ 2*(q1+q2)*tau1+2*tau2
  -(p1-p2)^2/(q1-q2)^2 ;
  simplify(series(Ltilde21-Ltilde21bis,p1=0));
```

$$Ltilde21bis := \lambda^3+(q1+q2)\,\lambda^2+\left(q1^2+q2\,q1+q2^2+2\,\tau 1\right)\lambda+q1^3+q1^2\,q2+q1\,q2^2+q2^3+2\,(q1+q2)\,\tau 1+2\,\tau 2-\frac{(p1-p2)^2}{(q1-q2)^2}$$

$$0 \qquad \textbf{(1.11)}$$

```
> c0:=0:
  Acheck11:=unapply(simplify(Acheck[1,1]),alpha7,alpha6,alpha5,
  alpha4,alpha3,alpha2,alpha1):
  Acheck12:=unapply(simplify(Acheck[1,2]),alpha7,alpha6,alpha5,
  alpha4,alpha3,alpha2,alpha1):
  Acheck21:=unapply(simplify(Acheck[2,1]),alpha7,alpha6,alpha5,
  alpha4,alpha3,alpha2,alpha1):
  Acheck22:=unapply(simplify(Acheck[2,2]),alpha7,alpha6,alpha5,
  alpha4,alpha3,alpha2,alpha1):
  Acheck111:=factor(Acheck11(0,0,0,0,2,0,0));
```

```
Acheck112:=factor(Acheck12(0,0,0,0,2,0,0));
Acheck121:=factor(Acheck21(0,0,0,0,2,0,0));
Acheck122:=factor(Acheck22(0,0,0,0,2,0,0));
```

$$Acheck111 := \frac{2\,(p1 - p2)}{3\,(q1 - q2)}$$
$$Acheck112 := \frac{2\,\lambda}{3} - \frac{2\,q1}{3} - \frac{2\,q2}{3}$$
$$Acheck121 := \frac{2}{3}\,\lambda^2 + \frac{2}{3}\,q1\,\lambda + \frac{2}{3}\,q2\,\lambda + \frac{2}{3}\,q1^2 + \frac{2}{3}\,q2^2 + \frac{4}{3}\,\tau 1$$
$$Acheck122 := -\frac{2\,(p1 - p2)}{3\,(q1 - q2)} \qquad \textbf{(1.12)}$$

```
> Acheck111:=factor(Acheck11(0,0,0,0,0,0,2));
Acheck112:=factor(Acheck12(0,0,0,0,0,0,2));
Acheck121:=factor(Acheck21(0,0,0,0,0,0,2));
Acheck122:=factor(Acheck22(0,0,0,0,0,0,2));
```

$$Acheck111 := 0$$
$$Acheck112 := 2$$
$$Acheck121 := 4\,q1 + 2\,\lambda + 4\,q2$$
$$Acheck122 := 0 \qquad \textbf{(1.13)}$$